

estimates for — Repeat postestimation command across models

[Description](#)
[Also see](#)[Quick start](#)[Syntax](#)[Options](#)[Remarks and examples](#)

Description

`estimates for` performs *postestimation_command* on each estimation result specified.

Quick start

Test for no effect of continuous covariate `x1` in stored estimates `m1` and `m2`

```
estimates for m1 m2: test x1==0
```

As above, but test interaction of binary covariate `a` and `x1`

```
estimates for m1 m2: test 0.a#c.x1==1.a#c.x1
```

Linear combination of coefficients of `x1` and `x2` in all stored estimates

```
estimates for _all: lincom x1 + x2
```

Tables of margins for each level of `a` and confidence intervals using estimates `m1` and `m2`

```
estimates for m1 m2: pwcompare i.a, cimargins
```

Syntax

`estimates for namelist [, options]: postestimation_command`

where *namelist* is a name, a list of names, `_all`, or `*`. A name may be `.`, meaning the current (active) estimates. `_all` and `*` mean the same thing.

<i>options</i>	Description
<code>noheader</code>	do not display title
<code>nostop</code>	do not stop if command fails

Options

`noheader` suppresses the display of the header as *postestimation_command* is executed each time.

`nostop` specifies that execution of *postestimation_command* is to be performed on the remaining models even if it fails on some.

Remarks and examples

[stata.com](http://www.stata.com)

In the example that follows, we fit a model two different ways, store the results, and then use `estimates for` to perform the same test on both of them:

► Example 1

```
. use http://www.stata-press.com/data/r14/auto
(1978 Automobile Data)
. generate gpm = 1/mpg
. regress gpm i.foreign i.foreign#c.weight displ
(output omitted)
. estimates store reg
. qreg gpm i.foreign i.foreign#c.weight displ
(output omitted)
. estimates store qreg
. estimates for reg qreg: test 0.foreign#c.weight==1.foreign#c.weight
```

Model **reg**

```
( 1) Ob.foreign#c.weight - 1.foreign#c.weight = 0
      F( 1, 69) = 4.87
      Prob > F = 0.0307
```

Model **qreg**

```
( 1) Ob.foreign#c.weight - 1.foreign#c.weight = 0
      F( 1, 69) = 0.03
      Prob > F = 0.8554
```

Also see

[R] [estimates](#) — Save and manipulate estimation results